
Answering questions at approx 15.00 (or by phone)

Functional Programming TDA 452, DIT 142

2017-04-11 14.00 – 18.00 “Maskin”-salar (M)

• There are 6 questions with maximum 6 + 8 + 6 + 5 + 5 + 10 = 40 points. Grading:

Chalmers: 3 = 20–26 points, 4 = 27–33 points , 5 = 34–40 points
GU: G = 20–33 points, VG = 34–40 points

• Results: latest approximately 10 days.

• Permitted materials:

– Dictionary

• Please read the following guidelines carefully:

– Read through all Questions before you start working on the answers.

– Begin each Question on a new sheet.

– Write clearly; unreadable = wrong!

– For each part Question, if your solution consists of more than a few lines of Haskell
code, use your common sense to decide whether to include a short comment to
explain your solution.

– You can use any of the standard Haskell functions listed at the back of this exam
document.

– Full points are given to solutions which are short, elegant, and correct. Fewer
points may be given to solutions which are unnecessarily complicated or unstruc-
tured.

– You are encouraged to use the solution to an earlier part of a Question to help
solve a later part — even if you did not succeed in solving the earlier part.

1. (6 points) Given the following type for trees,

data Tree a = Leaf a | Branch (Tree a) (Tree a) deriving (Eq,Show)

t = Branch (Leaf 5) (Branch (Leaf 1) (Leaf 10)) -- an example tree

define

• a function that computes the sum of the values in a tree, and

• a function that finds the minimum element in a tree.

sumTree :: Num a => Tree a -> a -- Example: sumTree t == 16

minTree :: Ord a => Tree a -> a -- Example: minTree t == 1

While these functions can be defined as two separate recursive functions, they will be
very similar, so this solution will give you at most 4 points. For full points, define a
(recursive) helper function that captures the common pattern, and make sumTree and
minTree two simple (non-recursive) functions that call the helper function.

Solution:

foldTree :: (a->a->a) -> Tree a -> a

foldTree f (Leaf a) = a

foldTree f (Branch l r) = f (foldTree f l) (foldTree f r)

sumTree = foldTree (+)

minTree = foldTree min

-- Alternative solution:

sumTree’ (Leaf a) = a

sumTree’ (Branch l r) = sumTree’ l + sumTree’ r

minTree’ (Leaf a) = a

minTree’ (Branch l r) = min (minTree’ l) (minTree’ r)

-- Another alternative solution:

toList :: Tree a -> [a]

toList (Leaf a) = [a]

toList (Branch l r) = toList l ++ toList r

sumTree’’ :: Num a => Tree a -> a

minTree’’ :: Ord a => Tree a -> a

sumTree’’ = sum . toList

minTree’’ = minimum . toList

2. (8 points)

(a) (3 points) Reimplement the following function by using recursion directly, travers-
ing the argument list only once.

splitAt :: Int -> [a] -> ([a],[a])

splitAt n xs = (take n xs,drop n xs)

Solution:

splitAt :: Int -> [a] -> ([a],[a])

splitAt n xs | n<=0 = ([],xs)

splitAt _ [] = ([],[])

splitAt n (x:xs) = (x:ys,zs) where (ys,zs) = splitAt (n-1) xs

-- For testing, not part of the solution

prop_splitAt n xs = splitAt n xs == (take n xs,drop n xs)

(b) (3 points) Define a function chunksOf that splits a lists into chunks of the length
indicated by the first argument.

chunksOf :: Int -> [a] -> [[a]]

-- Example: chunksOf 4 [1..10] == [[1,2,3,4],[5,6,7,8],[9,10]]

Solution:

chunksOf n [] = []

chunksOf n xs = xs1:chunksOf n xs2 where (xs1,xs2) = splitAt n xs

(c) (2 points) Write a property that describes what the expected length of chunksOf n xs

is, depending on n and xs.

Solution:

prop_length_chunksOf n xs =

n>0 ==> length (chunksOf n xs) == (length xs+(n-1)) ‘div‘ n

3. (6 points) For each of the following functions, give the most general type, or write No
type if the definition is not type correct in Haskell.

fa x = [x,x]

fb x y = if x then y else -y

fc (f,g) (x,y) = (f x,g y)

Solution:

fa :: x -> [x]

fb :: Num a => Bool -> a -> a

fc :: (a->b,c->d) -> (a,c) -> (b,d)

4. (5 points) Given the following data types for playing cards and hands of playing cards,

data Card = ... deriving (Eq,Show) -- the details are unimportant

data Hand = Empty | Add Card Hand deriving (Eq,Show)

rCard :: Gen Card -- generates a random Card

define a QuickCheck test data generator

rHand :: Gen Hand

that generates a random hand of cards that contains at least one card, at most 6 cards,
and doesn’t contain the same card twice.

Solution:

rHand = do size <- choose (1,6)

cards <- vectorOf size rCard

return (toHand (nub cards))

where

toHand :: [Card] -> Hand

toHand [] = Empty

toHand (c:cs) = Add c (toHand cs)

5. (5 points) Consider the following imperative style function that prints a string inside a
frame of asterisks:

printFramed :: String -> IO ()

printFramed s = -- printFramed " Hello world "

do putStars (n+2) --

putStr "\n*" -- ***************

putStr s -- * Hello world *

putStr "*\n" -- ***************

putStars (n+2)

putStr "\n"

where

n = length s

putStars 0 = return ()

putStars n = do putStr "*"

putStars (n-1)

Write a shorter, functional style implementation that produces the same output:

printFramed s = putStr (framed s)

framed :: String -> String

framed s = ...

To keep the definition short, use appropriate predefined functions instead of your own
recursive functions.

Solution:

framed :: String -> String

framed s = stars ++ "\n*" ++ s ++ "*\n" ++ stars ++ "\n"

where

stars = replicate n ’*’

n = length s + 2

6. (10 points) Consider the following representation of a simple variant of XML:

type TagName = String -- A few lowercase letters ’a’..’z’

data XML = Text String -- Arbitrary text

| Elem TagName [XML] -- A tagged element <t>...</t>

deriving (Eq,Show)

(a) (3 points) Write a function

tableToXML :: [[String]] -> XML

that takes a table (represented as a list of rows of cells) and creates an HTML
table for it (assuming here that HTML is an instance of XML). Example:

tableToXML [["A","B"],["C","D"]] ==

Elem "table" [Elem "tr" [Elem "td" [Text "A"],Elem "td" [Text "B"]],

Elem "tr" [Elem "td" [Text "C"],Elem "td" [Text "D"]]]

Solution:

tableToXML rows = Elem "table" (map rowToXML rows)

where

rowToXML row = Elem "tr" (map cellToXML row)

cellToXML cell = Elem "td" [Text cell]

(b) (5 points) Write a function

showXML :: XML -> String

that outputs the textual representation of an XML document. Some characters in
the text in the XML document need to be treated specially: since < starts a tag,
any < in text should be output as &lt;, and any & should be output as &amp;.

Examples:

showXML (Elem "td" [Text "hello"]) == "<td>hello</td>"

showXML (Text "this & that") == "this &amp; that"

showXML (Elem "td" [Text "4 < 5"]) == "<td>4 &lt; 5</td>"

Solution:

showXML (Text s) = escape s

showXML (Elem tag xml) = start tag++concatMap showXML xml++end tag

start,end :: TagName -> String

start tag = "<"++tag++">"

end tag = start (’/’:tag)

escape :: String -> String

escape s = concatMap escape1 s

where

escape1 ’<’ = "&lt;"

escape1 ’&’ = "&amp;"

escape1 c = [c]

(c) (2 points) Use the two functions above to define a function that creates the HTML
code for a table containing values of an arbitrary type in the Show class:

renderTable :: Show a => [[a]] -> String

Example:

renderTable [[1,2],[3,4]] ==

"<table><tr><td>1</td><td>2</td></tr>"++

"<tr><td>3</td><td>4</td></tr></table>"

Solution:

renderTable = showXML . tableToXML . map (map show)

